Caso Práctico II

Gestión y planificación

Optimización no lineal

José Ignacio Escribano

Móstoles, 29 de diciembre de 2015

Índice de figuras

Índice de tablas

Índice

1. Introducción

La empresa ELEKTRASA acaba de adquirir tres calentadores que conforman un nuevo generador de energía. Para cada calentador se conoce su función de coste, que son las siguientes:

$$\begin{aligned} C_1(x) &= 0.07\log(x^{100}) \\ C_2(x) &= 0.002x^2 \\ C_3(x) &= 0.05x \end{aligned}$$

La variable x determina la producción de cada calentador medida en Megavatios/hora. El coste $C_i(x)$ de cada calentador $i = 1, 2, 3$ viene dado en miles de euros. El generador no puede producir más de 1 000 MW/h. Cada calentador tiene unos límites (inferior y superior) de producción, los cuales son:

Calentador 1:	$[1, 600]$
Calentador 2:	$[0, 600]$
Calentador 3:	$[0, 600]$

Se quiere minimizar el coste de generación de energía.

El modelo a optimizar es el siguiente:

$$\begin{aligned} &\text{min} 0.07\log(x_1^{100}) + 0.02^2 + 0.05x_3 \\ &\text{s.a.} x_1 + x_2 + x_3 - 1000 = 0 \\ &\quad 1 \leq x_1 \leq 600 \\ &\quad 0 \leq x_2 \leq 600 \\ &\quad 0 \leq x_3 \leq 600 \end{aligned}$$

Resolviendo este problema se obtiene que el coste mínimo se consigue produciendo 600 MW con el primer calentador, 12.5 MW con el segundo y 387.5 MW con el tercero.

2. Resolución de las cuestiones de evaluación

A continuación resolveremos las cuestiones de evaluación planteadas.

2.1. Cuestión 1

Una vez definidas las nuevas funciones de coste y los límites de producción de energía, el problema de optimización no lineal queda de la siguiente manera:

$$
\begin{array}{lr}
\min & 0.0578082\log(x_1^{100}) + 0.0016517x_2^2 + 0.0412916x_3 \\
\text{s.a.} & x_1 + x_2 + x_3 - 1000 = 0 \\
& 1 \leq x_1 \leq 700 \\
& 0 \leq x_2 \leq 500 \\
& 0 \leq x_3 \leq 500
\end{array}
$$

2.2. Cuestión 2

Modificamos de forma adecuada los valores de E1, E2, E3 y del vector ub, quedando de la siguiente manera:

```
// Se define la función objetivo, su gradiente y su Hessiano

function [f,g,H] = central(x);

E1=0.0578082;
E2=0.0016517;
E3=0.0412916;

f = log(x(1)^100)*E1+x(2)^2*E2+x(3)*E3;

g = zeros(3,1);
g(1) = 100/x(1)*E1;
g(2) = 2*x(2)*E2;
g(3) = E3;

H = zeros(3,3);
H(1,1) = -100/x(1)^2*E1;
H(2,2) = 2*E2;
H(3,3)=  0;

endfunction //Fin de la funcion central

// Se define la restriccion de igualdad y su Jacobiano

function [c,A] = demanda(x);

c = x(1,1)+x(2,1)+x(3,1)-1000;
A = [1 1 1];

endfunction //Fin de la funcion demanda
```

```
// Se implementa una función con el método de puntos interiores

function [x,f] = interior(x,lb,ub);

  // METODO DE PUNTOS INTERIORES

  // Variables duales de las cotas inferiores

  n=length(x);

  zlb=zeros(n,1);
  for i=1:n
   zlb(i)=1/(x(i) - lb(i));
  end;

  // Variables duales de las cotas superiores

  zub=zeros(n,1);
  for i=1:n
   zub(i)=1/(ub(i) - x(i));
  end;

  // Calculamos el valor inicial de los multiplicadores

  [f,g,H] = central(x);
  [c,A] = demanda(x);
  m=length(c);
  lam=A'\(g-zlb+zub);

  // Inicializamos el contador de iteraciones

  k=0;

  // Condiciones de complementariedad

  colb=zeros(n,1);
  for i=1:n
    colb(i)=(x(i) - lb(i))'*zlb(i);
  end;

  coub=zeros(n,1);
  for i=1:n
```

```
  coub(i)=(ub(i) - x(i))'*zub(i);
end;

// Inicializamos el parámetro de barrera mu

mu=1;

// Comienza el bucle principal de la funcion interior

while norm([(g-A'*lam-zlb+zub)' c' colb' coub'])>1.e-5

// Construimos la matriz del sistema

  Xlb=zeros(n,n);
  for i=1:n
    Xlb(i,i)=1/(x(i)-lb(i));
  end;
  Xub=zeros(n,n);
  for i=1:n
    Xub(i,i)=1/(ub(i)-x(i));
  end;
  Zlb=diag(zlb);
  Zub=diag(zub);
  Hd=H+Xlb*Zlb+Xub*Zub;

  // Hacemos la matriz H definida positiva

  [U,D]=spec(Hd);
  for i=1:n
    D(i,i)=max(abs(D(i,i)),1.e-3);
  end
  Hd=U*D*U';
  K=[Hd A'
     A 0];

  // Calculamos el lado derecho del sistema de ecuaciones

  der=[-g+mu*diag(Xlb)-mu*diag(Xub)+A'*lam
       -c];

  // Resolvemos el sistema de ecuaciones

  d=K\der;
```

```
// Calculamos las direcciones de movimiento de las variables primal

dx=d(1:n,1);
dl=-d(n+1:n+m,1);

// Direccion de movimiento de las variables duales

dzlb = mu*diag(Xlb)-zlb-Xlb*Zlb*dx;
dzub = mu*diag(Xub)-zub+Xub*Zub*dx;

// Calculamos la longitud de paso de las variables duales

tau=0.99995;
azlb=1;
for i=1:n'
 if dzlb(i)<0;
  azlb=min(tau*(-zlb(i))/dzlb(i),azlb);
 end;
end;

azub=1;
for i=1:n'
 if dzub(i)<0;
  azub=min(tau*(-zub(i))/dzub(i),azub);
 end;
end;
alfaz=min([azlb azub]);

// Actualizamos las variables duales

zlb=zlb+alfaz*dzlb;
zub=zub+alfaz*dzub;

// Calculamos la longitud de paso de las variables primales

alb=1;
aub=1;

for i=1:n
 if dx(i)<0;
   alb=min(tau*(lb(i)-x(i))/dx(i),alb);
 end;
```

```
end;

for i=1:n
 if dx(i)>0;
  aub=min(tau*(ub(i)-x(i))/dx(i),aub);
 end;
end;

alfax=min([alb aub]);

// Actualizamos las variables primales y los multiplicadores

x=x+alfax*dx;
lam=lam+alfax*dl;

// Actualizamos los valores del problema

[f,g,H]= central(x);
[c,A] = demanda(x);

// Actualizamos el parámetro barrera

mu=0.9*mu;

// Actualizamos el contador de iteraciones

k=k+1;

// Actualizamos las condiciones de complementariedad

for i=1:n
  colb(i)=(x(i) - lb(i))'*zlb(i);
end;
for i=1:n
  coub(i)=(ub(i) - x(i))'*zub(i);
end;

// Limitamos el numero de iteraciones a 1000

if k>1000 then
  disp('Demasiadas iteraciones')
  break
end
```

```
  end

endfunction //Fin de la funcion interior

// PROGRAMA PRINCIPAL

// Se define el vector de cotas inferiores de cada calentador

lb=[1 0 0]';

// Se define el vector de cotas superiores de cada calentador

ub=[700 500 500]';

// Se toma un punto inicial contenido entre las cotas

x=[300 200 300]';

// Resolvemos el problema

[x,f]=interior(x,lb,ub);

// Sacamos por pantalla la solución del problema

disp(f,'El coste óptimo del GW/hora (en miles de euros) es:')
disp(x(1),'Los MW/hora a producir por el primer calentador son:')
disp(x(2),'Los MW/hora a producir por el segundo calentador son:')
disp(x(3),'Los MW/hora a producir por el tercer calentador son:')

// FIN DEL PROGRAMA PRINCIPAL
```

y resolviendo con Scilab tenemos que la solución del problema es la siguiente:

```
El coste óptimo del GW/hora (en miles de euros) es:

    50.000033

 Los MW/hora a producir por el primer calentador son:

    699.99988

 Los MW/hora a producir por el segundo calentador son:
```

```
   12.499822

 Los MW/hora a producir por el tercer calentador son:

   287.5003
```

2.3. Cuestión 3

A tenor de los resultados devueltos por Scilab, tenemos que que el coste del GW/hora es de 50 000 euros. Se deben producir 700 MW/hora del primer calentador, 12.5 MW/hora del segundo y 287.5 del tercer calentador para minimizar la función de coste.

3. Conclusiones